/*
	Estimates the degree of selection bias into the Glassdoor dataset for
	Nigerian college graduates employed in Nigeria
	Output:	Figure C1j
			Selection_Nigeria.csv
*/

local seed "C:\Users\jsock\Dropbox\Research\GD\International"

local dataPath "`seed'/Data"
local inputPath "`seed'/InputData"
local figurePath "`seed'/Replication/Figures"
local tablePath "`seed'/Replication/Tables"
local estimatePath "`seed'/Replication/Estimates"
local tempPath "`seed'/Replication/TempData"

********************************************************
* Save exchange rates as dta file
********************************************************

clear 
set more off
set matsize 10000

insheet using "`inputPath'/Exchange_rates_2022.csv", comma 
 
keep country_glassdoor year ppp_xrat  

save `tempPath'/Exchange_rates.dta, replace

*--------------------------------
* Thresholds used for sample selection
*--------------------------------
.
scalar country_premia_thresh = 25

scalar selection_thresh = 25

scalar school_thresh = 25 

local currentCountry = "Nigeria"
local saveCountry = "Nigeria"

********************************************************
* Read in UK undergraduate earnings
********************************************************

clear 
set more off
set matsize 10000

insheet using "`inputPath'/`saveCountry'_universities_earnings.csv", comma

generate country_glassdoor = "`currentCountry'"

merge m:1 country_glassdoor year using `tempPath'/Exchange_rates.dta

keep if _merge == 3
drop _merge

generate undergrad_pay =  average_annual * ppp_xrat

sort school year

collapse (mean) undergrad_pay , by(school)

replace school = "Adekunle Ajasin University" if school == "Adekunle Ajasin University, Akungba"
replace school = "Ahmadu Bello University" if school == "Ahmadu Bello University, Zaria"
replace school = "Babcock University" if school == "Babcock University, Ilishan-Remo"
replace school = "Lagos State University" if school == "Lagos State University Ojo, Lagos"
replace school = "Nnamdi Azikiwe University" if school == "Nnamdi Azikiwe University, Awka"
replace school = "Obafemi Awolowo University" if school == "Obafemi Awolowo University, Ife-Ife"
replace school = "Olabisi Onabanjo University" if school == "Olabisi Onabanjo University Ago-Iwoye"
replace school = "Federal University of Agriculture, Abeokuta" if school == "University of Agriculture, Abeokuta"
replace school = "University of Nigeria" if school == "University of Nigeria, Nsukka"
replace school = "University of Port Harcourt" if school == "University of Port-Harcourt"

save `tempPath'/`saveCountry'_universities_earnings_merge.dta, replace

********************************************************
* Read in country gdp to get universitycountry gdp
********************************************************

clear 
set more off
set matsize 10000
set scheme s1mono

* Set path and load data 
insheet using "`dataPath'/Salaries_international_dataset_main.csv", comma
drop v1

drop if jobtitle == ""

drop metro shortname 
drop city basecurrency country_iso dateval  
drop gender
drop sectorname iscurrentjobflag 

generate national_rank_pct = national_rank / numberuniversities

*--------------------------------
* Exclude users that leave more than 10 reviews
*--------------------------------

sort fk_userid yearofsalary 

by fk_userid : gen userReviews = _N

drop if userReviews > 10

drop userReviews

*--------------------------------
* Convert with ppp exchange rates
*--------------------------------

generate logbase = ln(basesalary * ppp_xrat)

drop if logbase == .

*--------------------------------
* Exclude outliers in base pay
*--------------------------------

scalar scalingThresh = 10

generate realbase = basesalary * ppp_xrat
generate outside_thresh =  (realbase < (1/scalingThresh) * gdppw) | (realbase > scalingThresh * gdppw) 
drop realbase

drop if outside_thresh

*--------------------------------
* Keep only Australia
*--------------------------------

drop if employertypecode == "SELF_EMPLOYED" 

keep if countryname == "`currentCountry'"

keep if universitycountry == "`currentCountry'"

*--------------------------------
* Generate indicators for valid in education analysis
*--------------------------------

generate hasDegree = degree != "UNMATCHED" & degree != "missing" & degree != ""  & degree != "HIGHSCHOOL"

generate uniDegree = degree == "BACHELORS" 

generate hasSchool = school != ""

generate valid_educ = uniDegree & hasDegree & hasSchool & universitycountry != ""

keep if valid_educ 

********************************************************
* Create valid sampels for one and two year comparisons
********************************************************

*--------------------------------
* Years since school ended
*--------------------------------

split endschool, p("-")

destring endschool1, gen(end_year)

generate years_from_graduate = yearofsalary - end_year

*--------------------------------
* Years since school ended
*--------------------------------

bys school : egen avg_grad = median(logbase) if inrange(years_from_graduate,1,5)

bys school : gen n_grad = sum(inrange(years_from_graduate,1,5))

*-------------------------------- 
* Add comparison earnings
*--------------------------------

local seed "C:\Users\jsock\Dropbox\Research\GD\International\InputData"

merge m:1 school using `tempPath'/`saveCountry'_universities_earnings_merge.dta

keep if _merge == 3
drop _merge

*--------------------------------
* Compare External data and GD data for recent graduates
*--------------------------------

preserve

	collapse (max) avg_grad n_grad undergrad_pay national_rank_pct, by(school)
	
	generate avg_median_undergrad = ln(undergrad_pay)
		
	* Difference between scorecard and gd 
	generate diff_grad = avg_grad - avg_median_undergrad	
	
	egen total_grad = sum(n_grad)
	
	keep if diff_grad != .
	
	* Get averages
	sum diff_grad
	local avg_equal = r(mean)
	scalar est_wtd_equal = r(mean)
	
	sum diff_grad [aw=n_grad]
	local avg_wtd_gd = r(mean)
	scalar est_wtd_gd = r(mean)
	scalar n_wt_wtd_gd = r(N)
	scalar sum_wt_wtd_gd = r(sum_w)

	* Get averages for top 5%
	sum diff_grad if national_rank_pct <= 0.05 [aw=n_grad]
	local avg_wtd_gd_top5 = r(mean)
	scalar est_wtd_gd_top5 = r(mean)
	scalar n_wt_wtd_gd_top5 = r(N)
	scalar sum_wt_wtd_gd_top5 = r(sum_w)
	
	* Get averages for not top 5%
	sum diff_grad if national_rank_pct > 0.05 | national_rank_pct == . [aw=n_grad]
	local avg_wtd_gd_not5 = r(mean)
	scalar est_wtd_gd_not5 = r(mean)
	scalar n_wt_wtd_gd_not5 = r(N)
	scalar sum_wt_wtd_gd_not5 = r(sum_w)

	* Weighted and unweighted kernel density
	twoway kdensity diff_grad [aw=n_grad], lcolor(gs3) lpattern(dash) xscale(r(-0.8(0.2)0.8)) xlabel(-0.8(0.2)0.8)  ///
		xline(0, lcolor(black) lpattern(solid) lwidth(thin)) fcolor(none) ///
		xtitle("log difference within college") ytitle("density") title("") scale(1.2)
	graph export "`figurePath'/Figure_C1j.eps"	, replace

restore

*--------------------------------
* Export summary of selection estimates
*--------------------------------

preserve

	generate selection_est_equal = est_wtd_equal
	generate selection_est_wtd = est_wtd_gd
	generate selection_est_wtd_top5 = est_wtd_gd_top5
	generate selection_est_wtd_not5 = est_wtd_gd_not5

	generate selection_wtd_N = n_wt_wtd_gd
	generate selection_wtd_N_top5 = n_wt_wtd_gd_top5
	generate selection_wtd_N_not5 = n_wt_wtd_gd_not5

	generate selection_wtd_sum = sum_wt_wtd_gd
	generate selection_wtd_sum_top5 = sum_wt_wtd_gd_top5
	generate selection_wtd_sum_not5 = sum_wt_wtd_gd_not5

	keep universitycountry selection_*
	
	keep if _n == 1
	
	order universitycountry selection_wtd_N* selection_wtd_sum* selection_est_* 
	
	outsheet using "`estimatePath'\Selection_`saveCountry'.csv" , comma replace

restore
